
Geometric data and materials: 
 
See slide. 
 
 
Numerical Solution 
 
BARI10 
 
The mesh generation file is: 
 
*%siz 40 
 
opti echo 1; 
* 
opti titr 'BARI - 10'; 
opti dime 2 elem qua4; 
* 
p0=0 0; 
p1=3.2E-3 0; 
p2=0 32.4E-3; 
tol=0.01E-3; 
* 
base=p0 d 6 p1; 
stru=base TRAN 60 p2; 
elim tol (stru et p2); 

* 
symax=stru poin droi p0 p2 tol; 
top=stru poin droi p2 (p1 PLUS p2) tol; 
* 
str1=chan 'POI1' stru; 
bas1=chan 'POI1' base; 
viti=diff str1 bas1; 
lili=diff bas1 (p0 et p1); 
slin=stru poin droi p1 (p1 plus p2) tol; 
* 
mesh=stru et symax et viti et lili et slin; 
tass mesh; 
* 
opti sauv form 'bari10.msh'; 
sauv form mesh; 

 
The input file is: 
 
BARI - 10 
*----------------------------------------------------------------------- 
ECHO 
 CONV win 
CAST MESH 
*-----------------------------------------------------------Problem type 
AXIS NONL ALE 
*-----------------------------------------------------------Dimensioning 
DIME 
  PT2L 1000 Q41 360 
  TABL 1 5 
  ECRO 3960 
  NALE 4 NBLE 354 
  SLPC 1 SLPN 61 
TERM 
*---------------------------------------------------------------Geometry 
GEOM Q41 STRU TERM 
*--------------------------------------------------Geometric Complements 
COMP EPAI 1 TOUS 
*------------------------------------------------------------Grid motion 
GRIL LAGR LECT P0 P1 TOP TERM 
     ALE TOUS 
     LIGN BASE LECT P0 P1 TERM LIST LECT LILI TERM 
     SLIP EQUI LECT SLIN TERM 

     AUTO AUTR 
*----------------------------------------------------------Material data 
MATE VM23 RO 8930. YOUN 1.17D11 NU 0.35D0 ELAS 4.D8 
          TRAC 2 4.D8 3.418803D-03 1.004D11 1000.003418803 
          TOUS 
*----------------------------------------------------Boundary conditions 
LINK COUP 
     BLOQ 2 LECT BASE TERM 
     CONT SPLA NX 1 NY 0 LECT SYMAX TERM 
*-----------------------------------------------------Initial conditions 
INIT VITE 2 -227. LECT VITI TERM 
*----------------------------------------------------------------Outputs 
ECRI DEPL VITE TFRE 10.E-6 
          POIN LECT P1 TOP TERM 
     FICH K200 TFRE 10.E-6 POIN TOUS 
          VARI DEPL VITE ECRO ECRC LECT 2 7 TERM 
     fich alic temp FREQ 1 
                    poin lect P1 P2 term 
     fich alic tfre 1.e-6 
*----------------------------------------------------------------Options 
OPTI NOTE 
     CSTA 0.5 
     LOG 1 
*--------------------------------------------------Transient calculation 

CALC TINI 0. TEND 80.D-6 
*==============================================================ANIMATION 
PLAY 
CAME   1 EYE   4.47289E-02 -3.66604E-02  6.13469E-02 
!        Q     9.09576E-01  2.93578E-01  2.86788E-01 -6.50846E-02 
         VIEW -4.83497E-01  5.71394E-01 -6.63129E-01 
         RIGH  8.27033E-01  4.99905E-02 -5.59926E-01 
         UP    2.86788E-01  8.19152E-01  4.96732E-01 
         FOV   2.48819E+01 
SCEN GEOM NAVI FREE 
          LINE HEOU SSHA 
     ISO  FILL FIEL ECRO 7 SCAL USER PROG 4.0D8 PAS 0.2D8 6.6D8 TERM 
     TEXT ISCA 
     colo pape 
     LIMA ON 
titl tit1 'EUROPLEXUS (C) Animation' 
     tit2 'Taylor Bar Impact' 
     tit3 'Author: F. Casadei' 
sler cam1 1        nfra 20 
trac offs fich avi               nocl nfto 101 fps 10 kfre 10 comp -1  
                                                 AXIS 27 270 REND 
sler cam1 1        nfra  1 
TRAC OFFS FICH AVI CONT NOCL 
                                                 AXIS 27 270 REND 

FREQ 0 tfre 1.d-6 
GOTR LOOP 79  OFFS FICH AVI CONT NOCL 
                                                 AXIS 27 270 REND 
GOTR          OFFS FICH AVI CONT 
                                                 AXIS 27 270 REND 
ENDPLAY 
*=========================================================POST-TREATMENT 
SUIT 
Post-treatment 
ECHO 
RESU ALIC TEMP GARD PSCR 
SORT GRAP 
AXTE 1000.0 'Time [ms]'  
*------------------------------------------------------Curve definitions 
COUR 1 'dx_P1'  DEPL COMP 1  LECT  P1 TERM 
COUR 2 'dy_P2'  DEPL COMP 2  LECT  P2 TERM 
*------------------------------------------------------------------Plots 
trac 1 2 axes 1.0 'DISPL. [M]' yzer 
*--------------------------------------------------Results qualification 
QUAL DEPL COMP 1 lect  P1 term REFE  3.90999E-3 TOLE 5.E-3 
     DEPL COMP 2 lect  P2 term REFE -1.09017E-2 TOLE 5.E-3 
*======================================================================= 
FIN 

 
The resulting final deformed mesh with superposed current yield stress (a measure of 
plastic deformation) is: 
 

 
The displacements are: 
 

 
 


